# Maintainer: Alexey Pavlov <alexpux@gmail.com>

_realname=gnome-icon-theme-symbolic
pkgbase=mingw-w64-${_realname}
pkgname="${MINGW_PACKAGE_PREFIX}-${_realname}"
pkgver=3.12.0
pkgrel=1
pkgdesc="GNOME icon theme, symbolic icons (mingw-w64)"
arch=('any')
license=('GPL')
url="https://www.gnome.org"
depends=("${MINGW_PACKAGE_PREFIX}-gtk2"
        "${MINGW_PACKAGE_PREFIX}-icon-naming-utils"
        "${MINGW_PACKAGE_PREFIX}-hicolor-icon-theme"
        "${MINGW_PACKAGE_PREFIX}-librsvg")
makedepends=('intltool')
install=gnome-icon-theme-symbolic-${CARCH}.install
groups=("${MINGW_PACKAGE_PREFIX}-gnome")
source=(https://download.gnome.org/sources/${_realname}/${pkgver%.*}/${_realname}-$pkgver.tar.xz)
sha256sums=('851a4c9d8e8cb0000c9e5e78259ab8b8e67c5334e4250ebcc8dfdaa33520068b')

prepare() {
  cd ${srcdir}/${_realname}-${pkgver}
}

build() {
  cd ${srcdir}/${_realname}-${pkgver}
  ./configure \
    --prefix=${MINGW_PREFIX} \
    --build=${MINGW_CHOST} \
    --host=${MINGW_CHOST}
  make
}

package() {
  cd "${srcdir}/${_realname}-${pkgver}"
  make DESTDIR="$pkgdir" install
  mkdir -p ${pkgdir}${MINGW_PREFIX}/lib
  mv ${pkgdir}${MINGW_PREFIX}/share/pkgconfig ${pkgdir}${MINGW_PREFIX}/lib/
}
